
启动transducer：into/sequence/transducer/eduction

状态

early termination（take，drop-while）

⾃自⼰己编写



Why

(defn workflow [ds]
  (->> ds
       (dedupe)
       (map #(* % %))
       (filter #(= 0 (rem % 111)))
       (reduce +)))

(workflow dataset)

(def dataset 
  (vec (interleave 
             (range 10000)
             (range))))



Why

dedupe

map

filter

reduce

result

ds

(def workflow2
  (comp
   (dedupe)
   (map #(* % %))
   (filter #(= 0 (rem % 111)))))

(transduce workflow2 + dataset)



Why

输⼊入源不不是 collection 时，如何复⽤用转化函数？



Transducer 是什什么

Transducer 是⼀一个函数 

参数为 reducing function 

返回值为 reducing function



Reducing Function

(accumulateN-1, item) —-> accumulateN 

(conj [1 2] 3) ===> [1 2 3]



Transducer 是什什么

(fn [xf]
  (fn ([] ...)
      ([result] ...)
      ([result input] ...)))



Reduce/fold

⼀一类访问递归类型数据结构的函数

https://en.wikipedia.org/wiki/Fold_(higher-order_function)


为什什么需要 Transducer

性能 

no interim collections 

no extra boxes 

复⽤用 

collection/channel/Observable/Stream



性能⽐比较

(defn workflow [ds]
  (->> ds
       (dedupe)
       (map #(* % %))
       (filter #(= 0 (rem % 111)))
       (reduce +)))

(bench (workflow dataset))

(def workflow2
  (comp
   (dedupe)
   (map #(* % %))
   (filter #(= 0 (rem % 111)))))

(bench (transduce workflow2 + dataset))

(defonce dataset (vec (interleave (range 10000) (range))))

Execution time mean : 1.794328 ms Execution time mean : 1.143402ms

https://github.com/hugoduncan/criterium


复⽤用
(require '[clojure.core.async :refer [>! <! <!!] :as a])

(def xform (comp (filter odd?) (map inc)))

(defn process [items]
  (let [out (a/chan 1 xform)
        in (a/to-chan items)]
    (a/go-loop []
      (if-some [item (<! in)]
        (do
          (>! out item)
          (recur))
        (a/close! out)))
    (<!! (a/reduce conj [] out))))

(process (range 10))
;; [2 4 6 8 10]



1.7 版本，以下函数被重写，输⼊入⼀一个参数时返
回⼀一个 transducer 

map cat mapcat filter remove take take-
while take-nth drop drop-while replace 
partition-by partition-all keep keep-
indexed map-indexed distinct interpose 
dedupe random-sample…

使⽤用案例例



启动 transducer
(def nums (range 20))

(def xf (comp (filter even?)
              (map inc)))

(transduce xf conj nums)
(into [] xf nums)
;; [1 3 5 7 9 11 13 15 17 19]

(sequence xf nums)
(eduction xf nums)
;; (1 3 5 7 9 11 13 15 17 19)



transduce/into/sequence/eduction

transduce 与 reduce 类型， ⾮非惰性 

into 内部使⽤用 transduce 实现 

sequence，惰性，cache 结果 

eduction，惰性，没有 cache 结果，每次计算



(def cnt (atom 0))
(take 10 (transduce (map #(do (swap! cnt inc) %)) conj () (range 1000)))
;; (999 998 997 996 995 994 993 992 991 990)
@cnt
;; 1000

(def cnt1 (atom 0))
(let [res (eduction (map #(do (swap! cnt1 inc) %)) (range 10))]
  (conj (rest res) (first res))
  @cnt1)
;; 20

(def cnt2 (atom 0))
(let [res (sequence (map #(do (swap! cnt2 inc) %)) (range 10))]
  (conj (rest res) (first res)) ; (2)
  @cnt2)
;; 10

transduce/into/sequence/eduction



状态

(defn dedupe []
  (fn [xf]
    (let [prev (volatile! ::none)]
      (fn
        ([] (xf))
        ([result] (xf result))
        ([result input]
          (let [prior @prev]
            (vreset! prev input)
              (if (= prior input)
                result
                (xf result input))))))))



Early termination

reduced 返回⼀一个reduced值，表明reduction
结束 

reduced? 

deref/@  取出 reduced 包含的值



Early termination
(defn take [n]
  (fn [rf]
    (let [nv (volatile! n)]
      (fn
        ([] (rf))
        ([result] (rf result))
        ([result input]
         (let [n @nv
               nn (vswap! nv dec)
               result (if (pos? n)
                        (rf result input)
                        result)]
           (if (not (pos? nn))
             (ensure-reduced result)
             result)))))))



⾃自⼰己编写transducer

(defn template-transducer [xf]
  (fn 
    ;; SET-UP
    ([]              (xf))
    ;; TEAR-DOWN
    ([result]        (xf result))
    ;; PROCESS
    ([result input]  (xf result input))))



扩展

https://labs.uswitch.com/transducers-from-
the-ground-up-the-essence/ 

use transducer if possible

https://labs.uswitch.com/transducers-from-the-ground-up-the-essence/
https://labs.uswitch.com/transducers-from-the-ground-up-the-essence/

